

VERSION 8.1

Mplus LANGUAGE ADDENDUM

In this addendum, changes introduced in Version 8.1 are described. They include corrections to minor problems that have been found since the release of Version 8 in April 2017 as well as the following new features:

- Time series analysis of intensive longitudinal data using residual dynamic structural equation modeling (RDSEM: Asparouhov, Hamaker, & Muthén, 2018; Asparouhov & Muthén, 2018a)
- Latent variable decomposition (latent variable centering) of predictors with random slopes for DSEM, RDSEM, and other TYPE=TWOLEVEL models using the Bayes estimator (Asparouhov & Muthén, 2018b) as shown in Example 9.1 for a random intercept model in the Mplus Version 8 User's Guide
- Lagged categorical outcomes and random slopes for categorical predictors using latent variable decomposition (latent variable centering) for DSEM and other TYPE=TWOLEVEL models using the Bayes estimator (Asparouhov & Muthén, 2018b)
- Automatic checking of whether two structural equation models are nested or equivalent (Bentler & Satorra, 2010) implemented and expanded to multiple group models and the weighted least squares estimators (Asparouhov & Muthén, 2018c)
- Expanded and modified SRMR (Asparouhov & Muthén, 2018d)
- Bivariate residual tests and factor score standard errors for the weighted least squares estimators
- Cluster-specific plots for TWOLEVEL

- Expanded odds ratio output
- Standard errors for correlations with TYPE=BASIC using the H1SE option of the OUTPUT command

TIME SERIES ANALYSIS OF INTENSIVE LONGITUDINAL DATA USING RESIDUAL DYNAMIC STRUCTURAL EQUATION MODELING (RDSEM)

Time series analysis of intensive longitudinal data using dynamic structural equation modeling (DSEM; Asparouhov, Hamaker, & Muthén, 2018) focuses on the regression of an outcome at a certain timepoint on the same outcome at one or more previous time points. Residual dynamic structural equation modeling (RDSEM; Asparouhov, Hamaker, & Muthén, 2018; Asparouhov & Muthén, 2018a) focuses on a regression of an outcome at a certain time point on one or more predictors at the same time point. In RDSEM, the autoregression across time is specified for the residual of the outcome. RDSEM is available for both N=1 and multilevel models with two levels.

REGRESSION ANALYSIS

The following RDSEM example is based on the DSEM Example 9.31 in the Mplus Version 8 User's Guide. The only difference is the specification of the random slope γ in the MODEL command. In DSEM, the dependent variable y is regressed on the dependent variable y_{t-1} which is y at the previous time point. In RDSEM, the residual of the dependent variable y referred to as y^* is regressed on the residual of the dependent variable y at the previous time point referred to as y^*_{t-1} .

```

VARIABLE: NAMES = y x w xm subject;
          WITHIN = x;
          BETWEEN = w xm;
          CLUSTER = subject;
          LAGGED = y(1);
DEFINE:   CENTER X (GROUPMEAN);
ANALYSIS: TYPE = TWOLEVEL RANDOM;
          ESTIMATOR = BAYES;
          PROCESSORS = 2;
          BITERATIONS = (2000);
MODEL:    %WITHIN%
          sy | y^ ON y^1; !DSEM: sy | y ON y&1;
          sx | y ON x;
          logv | y;
          %BETWEEN%
          y ON w xm;
          sy ON w xm;
          sx ON w xm;
          logv ON w xm;
          y sy sx logv WITH y sy sx logv;

```

In the example above, there is no missing data on the covariate x . When there is missing data on the covariate x which often occurs in conjunction with the `TINTERVAL` option, any observation with missing on x at one or more time points is not used in the analysis. To avoid this, the covariate x can be brought into the model by specifying its autoregression on previous time points. It is then treated as a dependent variable in the analysis and missing data theory can be used. Following is an example of how to specify the autoregression of x on previous time points in the RDSEM context:

```
x^ ON x^1;
```

In the example above, the covariate x is on the `WITHIN` list and a cluster-level variable for x referred to as xm is on the `BETWEEN` list. The covariate x is group-mean centered in the `DEFINE` command. An alternative to using the observed covariates x and xm is to do a latent variable decomposition (latent variable centering) of x into latent within and between components and use those in the analysis (Asparouhov & Muthén, 2018b). The partial

input is shown below. The WITHIN option and DEFINE command are commented out and the cluster-level xm variable is not used. In the MODEL command, the covariate x refers to the latent within-part of x in the within part of the model and the latent between-part of x in the between part of the model.

```
VARIABLE: NAMES = y x w xm subject;
          USEVARIABLES = y x w;
!        WITHIN = x;
          BETWEEN = w;
          CLUSTER = subject;
          LAGGED = y(1);
!DEFINE: CENTER X (GROUPMEAN);
ANALYSIS: TYPE = TWOLEVEL RANDOM;
          ESTIMATOR = BAYES;
          PROCESSORS = 2;
          BITERATIONS = (2000);
MODEL:   %WITHIN%
          sy | y^ ON y^1;
          sx | y ON x;
          logv | y;
          %BETWEEN%
          y ON w x;
          sy ON w x;
          sx ON w x;
          logv ON w x;
          w WITH x;
          y sy sx logv WITH y sy sx logv;
```

FACTOR ANALYSIS

The following RDSEM example is based on the DSEM Example 9.34 in the Mplus Version 8 User's Guide. There are four differences: the addition of a covariate x, a random slope regression of the factor f on the covariate x, the specification of an autoregression for the residual of the factor f, and the specification of the autoregressions for the residuals of the factor indicators y1 through y4. There is a latent variable decomposition (latent variable centering) of the covariate x.

```

VARIABLE: NAMES = y1-y4 x subject;
          CLUSTER = subject;
          LAGGED = y1-y4 (1);
ANALYSIS: TYPE = TWOLEVEL RANDOM;
          ESTIMATOR = BAYES;
          PROCESSORS = 2;
          BITERATIONS = (2000);
MODEL:   %WITHIN%
          f BY y1-y4(&1);
          s | f ON x;
          f^ ON f^1;
          y1^-y4^ PON y1^1 - y4^1;
          logv | f;
          %BETWEEN%
          fb BY y1-y4*;
          fb@1;
          fb s logv ON x;
          fb s logv WITH fb s logv;

```

LINEAR TREND ANALYSIS

The following RDSEM example is based on the DSEM Example 9.37 in the Mplus Version 8 User's Guide. The only difference is the specification of the random slope s_y in the MODEL command. In RDSEM, the residual of the dependent variable y referred to as y^{\wedge} is regressed on the residual of the dependent variable y at the previous time point referred to as $y^{\wedge 1}$.

```

VARIABLE: NAMES = y x w xm time subject;
          WITHIN = x time;
          BETWEEN = w xm;
          CLUSTER = subject;
          LAGGED = y(1);
DEFINE:   CENTER x (GROUPMEAN);
ANALYSIS: TYPE = TWOLEVEL RANDOM;
          ESTIMATOR = BAYES;
          PROCESSORS = 2;
          BITERATIONS = (10000);
MODEL:    %WITHIN%
          sy | y^ ON y^1;
          sx | y ON x;
          s | y ON time;
          logv | y;
          %BETWEEN%
          sy ON w xm;
          sx ON w xm;
          s ON w xm;
          logv ON w xm;
          y ON w xm;
          sy-logv y WITH sy-logv y;

```

LAGGED CATEGORICAL OUTCOMES AND RANDOM SLOPES FOR CATEGORICAL PREDICTORS USING LATENT VARIABLE DECOMPOSITION (LATENT VARIABLE CENTERING) FOR DSEM AND OTHER TYPE=TWOLEVEL MODELING USING THE BAYES ESTIMATOR

The following DSEM example shows a model with a lagged categorical outcome and a categorical predictor with a random slope using a latent variable decomposition (latent variable centering) of both the outcome and the predictor (Asparouhov & Muthén, 2018b).

```

VARIABLE: NAMES = y x w xm subject;
USEVARIABLES = y x w;
CATEGORICAL = y x;
BETWEEN = w;
CLUSTER = subject;
LAGGED = y(1);
ANALYSIS: TYPE = TWOLEVEL RANDOM;
ESTIMATOR = BAYES;
PROCESSORS = 2;
BITERATIONS = (2000);
MODEL: %WITHIN%
sy | y ON y&1;
sx | y ON x;
%BETWEEN%
y ON w x;
sy ON w x;
sx ON w x;
w WITH x;
y sy sx WITH y sy sx;

```

AUTOMATIC CHECKING OF WHETHER TWO STRUCTURAL EQUATION MODELS ARE NESTED OR EQUIVALENT

Automatic checking of whether two structural equation models are nested or equivalent (Bentler & Satorra, 2010) has been implemented and expanded to multiple group models and the weighted least squares estimators (Asparouhov & Muthén, 2018c). This checking is done in two steps. In the first step for nested models, the H0 model (the most restricted model with fewer parameters) is run using the NESTED option of the SAVEDATA command.

```

SAVEDATA:
NESTED IS nested.dat;

```

where nested.dat is the file where the model estimated means, variances and covariances are saved for use as sample statistics in the second step.

In the second step for nested models, the H1 model (the least restricted model with more parameters) is run using the NESTED option of the ANALYSIS command.

ANALYSIS:
NESTED IS nested.dat;

For equivalent models with the same number of parameters, the order does not matter.

The results show if the models are nested, not nested, equivalent, or no conclusion can be reached.

REFERENCES

Asparouhov, T., Hamaker, E.L. & Muthén, B. (2018). Dynamic structural equation models. *Structural Equation Modeling: A Multidisciplinary Journal*, 25:3, 359-388.

Asparouhov, T. & Muthén, B. (2018a). Comparison of DSEM and RDSEM. Technical appendix. Los Angeles: Muthén & Muthén.

Asparouhov, T. & Muthén, B. (2018b). Latent variable centering of predictor and mediator variables in multilevel and time series models. Technical appendix. Los Angeles: Muthén & Muthén.

Asparouhov, T. & Muthén, B. (2018c). Nesting and equivalence testing in Mplus. Technical appendix. Los Angeles: Muthén & Muthén.

Asparouhov, T. & Muthén, B. (2018d). SRMR in Mplus. Technical appendix. Los Angeles: Muthén & Muthén.

Bentler, P. & Satorra, A. (2010). Testing model nesting and equivalence. *Psychological Methods*, 15, 111-123.